Blatt 13

Aufgabe 50

```
> restart:
> dgl := (x^2 * y(x) + y(x)^3) * diff(y(x), x) + x^3 + x * y(x)^2
  = 0;
> aw := y(1) = y1;
```

$$dgl := \left(x^2\, y(x) + y(x)^3\right) y'(x) + x^3 + x\, y(x)^2 = 0$$

$$aw := y(1) = y1 \qquad \textbf{(1.1)}$$

Lösungen für y0 = 1.

```
> y1 := 1;
> lsg := dsolve({ dgl, aw }, y(x));
```

$$y1 := 1$$

$$lsg := y(x) = -\mathrm{I}\, x,\ y(x) = \mathrm{I}\, x,\ y(x) = \sqrt{-x^2 + 2} \qquad \textbf{(1.2)}$$

```
> # Die beiden komplexen Lösungen erfüllen die Anfangsbedingung
  nicht!
> is(subs(x = 1, rhs(lsg[1])) - y1 = 0);
> is(subs(x = 1, rhs(lsg[2])) - y1 = 0);
> is(subs(x = 1, rhs(lsg[3])) - y1 = 0);
```

false

false

true **(1.3)**

```
> u1 := unapply(rhs(lsg[3]), x);
```

$$u1 := x \mapsto \sqrt{-x^2 + 2} \qquad \textbf{(1.4)}$$

```
> # Definitionsbereich -sqrt(2)..sqrt(2).
> plot(u1(x), x = -sqrt(2)..sqrt(2), color = red, legend = u1,
  scaling = constrained);
```



Lösungen für y0 = 0.

```
> y1 := 0;
> lsg := dsolve({ dgl, aw }, y(x));
```

$$y1 := 0$$

$$lsg := y(x) = -\mathrm{I}\,x,\ y(x) = \mathrm{I}\,x,\ y(x) = \sqrt{-x^2+1},\ y(x) = -\sqrt{-x^2+1} \qquad \textbf{(1.5)}$$

```
> # Die beiden komplexen Lösungen erfüllen die Anfangsbedingung
  nicht!
> seq(print(rhs(l), subs(x = 1, rhs(l)), is(subs(x = 1, rhs(l) -
  y1 = 0))), l in lsg);
```

$$-\mathrm{I}\,x,\ -\mathrm{I},\ false$$

$$\mathrm{I}\,x,\ \mathrm{I},\ false$$

$$\sqrt{-x^2+1},\ 0,\ true$$

$$-\sqrt{-x^2+1},\ 0,\ true \qquad \textbf{(1.6)}$$

```
> u2 := unapply(rhs(lsg[3]), x);
> u3 := unapply(rhs(lsg[4]), x);
```

$$u2 := x \mapsto \sqrt{-x^2+1}$$

$$u3 := x \mapsto -\sqrt{-x^2+1} \quad (1.7)$$

```
> # Definitionsbereich -1..1 für beide reellen Lösungen.
> plot([ u2(x), u3(x) ], x = -1..1, color = [ red, blue ], legend
  = [ u2, u3 ], scaling = constrained);
```




```
> # Lösungen in x = 1 diff'bar?
> subs(x = 1, diff(u2(x), x));
> subs(x = 1, diff(u3(x), x));
Error, numeric exception: division by zero
Error, numeric exception: division by zero
> # Nein!
```

Aufgabe 51

```
> restart:
> dgl := diff(y(x), x) - y(x)/x - x^2 - x^3 + x^4 = 0;
```

$$dgl := y'(x) - \frac{y(x)}{x} - x^2 - x^3 + x^4 = 0 \quad (2.1)$$

```
> # Alle Lösungen der DGL
```

```
> lsg_allg := dsolve(dgl, y(x));
> lsg_allg := rhs(dsolve(dgl, y(x))):
```

$$lsg\_allg := y(x) = x\_C1 - \frac{x^3 (3x^2 - 4x - 6)}{12} \quad (2.2)$$

```
> aw := y(1) = y1;
> y1s := 1, 0, -1;
```

$$aw := y(1) = y1$$

$$y1s := 1, 0, -1 \quad (2.3)$$

```
> for y1 in y1s do
    'y1' = y1;
    lsg[y1] := rhs(dsolve({ dgl, aw }, y(x)));
    # Wahlweise:
    C := solve({ subs(x = 1, lsg_allg) = y1 }, { _C1 });
    lsg2[y1] := subs(C[1], lsg_allg);
    lsg[y1] - lsg2[y1];
  end do;
```

$$y1 = 1$$

$$lsg_1 := \frac{5x}{12} - \frac{x^3 (3x^2 - 4x - 6)}{12}$$

$$C := \left\{ \_C1 = \frac{5}{12} \right\}$$

$$lsg2_1 := \frac{5x}{12} - \frac{x^3 (3x^2 - 4x - 6)}{12}$$

$$0$$

$$y1 = 0$$

$$lsg_0 := -\frac{7x}{12} - \frac{x^3 (3x^2 - 4x - 6)}{12}$$

$$C := \left\{ \_C1 = -\frac{7}{12} \right\}$$

$$lsg2_0 := -\frac{7x}{12} - \frac{x^3 (3x^2 - 4x - 6)}{12}$$

$$0$$

$$y1 = -1$$

$$lsg_{-1} := -\frac{19x}{12} - \frac{x^3 (3x^2 - 4x - 6)}{12}$$

$$C := \left\{ \_C1 = -\frac{19}{12} \right\}$$

$$lsg2_{-1} := -\frac{19x}{12} - \frac{x^3 (3x^2 - 4x - 6)}{12}$$

$$0 \quad (2.4)$$

```
> plot([ seq(lsg[y1], y1 in y1s) ], x = -1.5..2.5);
```




```
> # Die Graphen schneiden sich in der Null. Der Satz von
  # Picard-Lindelöff setzt aber Lipschitz-Stetigkeit im 2.
  # Argument voraus, um (lokale) Eindeutigkeit zu
  # garantieren. Dies ist in der Null nicht erfüllt.
```

Aufgabe 52

```
> restart:
> dgl := diff(y(x), x) = sqrt(x * y(x));
> aw := y(1) = 4;
```

$$dgl := y'(x) = \sqrt{x\,y(x)}$$

$$aw := y(1) = 4 \tag{3.1}$$

```
> lsg := dsolve({ dgl, aw }, y(x));
```

$$lsg := y(x) = -\frac{x^3}{9} + \frac{2\,x\left(x^2 + 5\sqrt{x}\right)}{9} + \frac{25}{9} \tag{3.2}$$

```
> u1 := x -> rhs(lsg);
```

$$u1 := x \mapsto rhs(lsg) \tag{3.3}$$

```
> test := simplify(subs(y(x) = u1(x), dgl));
> simplify(rhs(test) - lhs(test)) assuming x > 0;
```

$$test := \frac{x^2}{3} + \frac{5\sqrt{x}}{3} = \frac{\sqrt{(x^{3/2}+5)^2 x}}{3}$$

$$0 \tag{3.4}$$

```
> # Aha!
> a := x -> 0;
> b := x -> -sqrt(x);
> s := 1/2;
```

$$a := x \mapsto 0$$

$$b := x \mapsto -\sqrt{x}$$

$$s := \frac{1}{2} \tag{3.5}$$

```
> dgl_neu := diff(y(x), x) + a(x) * y(x) + b(x) * y(x)^s = 0;
> # Test
> simplify(lhs(dgl_neu) - rhs(dgl_neu) - (lhs(dgl) - rhs(dgl)))
  assuming x > 0;
```

$$dgl\_neu := y'(x) - \sqrt{x}\sqrt{y(x)} = 0$$

$$0 \tag{3.6}$$

```
> # Korrespondierende DGL
> korr_dgl := diff(w(x), x) + (1 - s)*a(s)*w(x) + (1 - s)*b(x) =
  0;
> # Brauchen f^2(1) = y(1) = 4, also
> korr_aw := w(1) = sqrt(rhs(aw));
```

$$korr\_dgl := w'(x) - \frac{\sqrt{x}}{2} = 0$$

$$korr\_aw := w(1) = 2 \tag{3.7}$$

```
> f := rhs(dsolve({ korr_dgl, korr_aw }, w(x)));
> 'f^2' = expand(f^2);
```

$$f := \frac{x^{3/2}}{3} + \frac{5}{3}$$

$$f^2 = \frac{x^3}{9} + \frac{10\,x^{3/2}}{9} + \frac{25}{9} \tag{3.8}$$

```
> # Teste wieder die DGL:
> test2 := simplify(subs(y(x) = f^2, dgl));
```

$$test2 := \frac{(x^{3/2}+5)\sqrt{x}}{3} = \frac{\sqrt{(x^{3/2}+5)^2 x}}{3} \tag{3.9}$$

```
> simplify(rhs(test2) - lhs(test2)) assuming x > 0;
```

$$0 \tag{3.10}$$

Aufgabe 53

```
> restart:
> with(plots):
> dgl := diff(y(x), x) + y(x) * sin(x) + sin(2*x) = 0;
```

$$dgl := y'(x) + y(x)\sin(x) + \sin(2x) = 0 \quad (4.1)$$

```
> aw := y(0) = y0;
```

$$aw := y(0) = y0 \quad (4.2)$$

```
> y0s := [ 2, 1, 0 ];
```

$$y0s := [2, 1, 0] \quad (4.3)$$

```
> lsg := dsolve({ dgl, aw }, y(x));
```

$$lsg := y(x) = -2\cos(x) - 2 - \frac{e^{\cos(x)}(-y0 - 4)}{e} \quad (4.4)$$

```
> lsg2 := unapply(rhs(lsg), (x, y0));
```

$$lsg2 := (x, y0) \mapsto -2\cos(x) - 2 - \frac{e^{\cos(x)}(-y0 - 4)}{e} \quad (4.5)$$

```
> us := map(Y0 -> (x -> lsg2(x, Y0)), y0s);
```

$$us := [x \mapsto lsg2(x, 2), x \mapsto lsg2(x, 1), x \mapsto lsg2(x, 0)] \quad (4.6)$$

```
> # Richtungsfeld der DGL
> v := <1, rhs(isolate(dgl, diff(y(x), x)))>;
> # Vektorfeld normieren
> N := sqrt(v[1]^2 + v[2]^2);
> w := v/N;
```

$$v := \begin{bmatrix} 1 \\ -y(x)\sin(x) - \sin(2x) \end{bmatrix}$$

$$N := \sqrt{1 + (-y(x)\sin(x) - \sin(2x))^2}$$

$$w := \begin{bmatrix} \frac{1}{\sqrt{1 + (-y(x)\sin(x) - \sin(2x))^2}} \\ \frac{-y(x)\sin(x) - \sin(2x)}{\sqrt{1 + (-y(x)\sin(x) - \sin(2x))^2}} \end{bmatrix} \quad (4.7)$$

```
> p_field := fieldplot(w, x = -Pi .. 2*Pi, y = -1 .. 3, color =
  blue):
> p_sol := plot([ seq(u(x), u in us) ],
    x = -Pi .. 2*Pi, color = [ red, green, yellow ]):
> display({ p_field, p_sol });
```



Aufgabe 54

```
> restart:
> with(plots):
> with(VectorCalculus):
> with(LinearAlgebra):
> BasisFormat(false):
> f := (x, y) -> x^2 * (y + 1) + y/2;
```

$$f := (x, y) \mapsto x^2 (y+1) + \left(y \cdot \left(\frac{1}{2}\right)\right) \quad (5.1)$$

```
> g := (x, y) -> x^2 + y^2 - 1;
```

$$g := (x, y) \mapsto x^2 + y^2 - 1 \quad (5.2)$$

```
> # Parametrisierung
> param := t -> ( cos(t), sin(t) );
```

$$param := t \mapsto (\cos(t), \sin(t)) \quad (5.3)$$

```
> (f@param)(t);
```

(5.4)

$$\cos(t)^2\,(\sin(t)+1)+\frac{\sin(t)}{2} \tag{5.4}$$

```
> # Ansehen
> pf := plot3d(f(x, y), x = -1.1..1.1, y = -1.1..1.1):
> pf_constr := spacecurve([ param(t)[1], param(t)[2], (f@param)
  (t), t = -Pi..Pi ], x = -1..1, y = -1..1, thickness = 3, color
  = black ):
> display([ pf, pf_constr ]);
```




```
> L := (x, y, lambda) -> f(x, y) + lambda * g(x, y);
```

$$L := (x, y, \lambda) \mapsto f(x, y) + (\lambda\, g(x, y)) \tag{5.5}$$

```
> dL := (x, y, lambda) -> < seq(D[ll](L)(x, y, lambda), ll = 1.
  .3) >;
```

$$dL := (x, y, \lambda) \mapsto \langle seq(VectorCalculus{:-}\mathrm{D}_{ll}(L)(x, y, \lambda),\ ll = 1\,..3)\rangle \tag{5.6}$$

```
> d2L := (x, y, lambda) -> <seq(
     Transpose(< seq(D[ll,kk](L)(x, y, lambda), ll = 1..3) >),
    kk = 1..3) >;
```

$$d2L := (x, y, \lambda) \mapsto \langle seq(\langle seq(VectorCalculus{:-}\mathrm{D}_{ll,\,kk}(L)(x, y, \lambda),\ ll = 1\,..3)\rangle^{+}, \tag{5.7}$$

$kk = 1..3) \rangle$

```
> #L(1,1,1), dL(1,1,1), d2L(1,1,1);
> kritische_punkte := seq(allvalues(s),
    s in solve({ seq(dL(x, y, lambda)[ll] = 0, ll = 1..3) },
      { x, y, lambda }));
```

$$kritische\_punkte := \left\{\lambda = -\frac{1}{4}, x = 0, y = 1\right\}, \left\{\lambda = \frac{1}{4}, x = 0, y = -1\right\}, \left\{\lambda = -\frac{2}{3} - \frac{\sqrt{22}}{6}, x = \frac{\sqrt{10+4\sqrt{22}}}{6}, y = -\frac{1}{3} + \frac{\sqrt{22}}{6}\right\}, \left\{\lambda = -\frac{2}{3} - \frac{\sqrt{22}}{6}, x = -\frac{\sqrt{10+4\sqrt{22}}}{6}, y = -\frac{1}{3} + \frac{\sqrt{22}}{6}\right\}, \left\{\lambda = -\frac{2}{3} + \frac{\sqrt{22}}{6}, x = \frac{I}{6}\sqrt{-10+4\sqrt{22}}, y = -\frac{1}{3} - \frac{\sqrt{22}}{6}\right\}, \left\{\lambda = -\frac{2}{3} + \frac{\sqrt{22}}{6}, x = -\frac{I}{6}\sqrt{-10+4\sqrt{22}}, y = -\frac{1}{3} - \frac{\sqrt{22}}{6}\right\} \quad (5.8)$$

```
> for kr in kritische_punkte do
    #kr;
    print('<x, y>' = subs(kr, [x, y]));
    if (not is(subs(kr, x), real)) or (not is(subs(kr, y), real))
  then
     # Komplexe Werte überspringen
      print("Komplexe kritische Stelle ignoriert");
       next;
    end if;
    #'<x, y>' = evalf(subs(kr, [x, y]));
    print('f(x, y)' = subs(kr, f(x, y)));
   # Hessematrix
   d2L_val := subs(kr, d2L(x, y, lambda)):
   print(Re(evalf(simplify(Eigenvalues(d2L_val)))));
  end do:
```

$$\langle x, y \rangle = [0, 1]$$

$$f(x, y) = \frac{1}{2}$$

$$\begin{bmatrix} 3.500000000 \\ -2.265564437 \\ 1.765564437 \end{bmatrix}$$

$$\langle x, y \rangle = [0, -1]$$

$$f(x, y) = -\frac{1}{2}$$

$$\begin{bmatrix} 0.5000000000 \\ -1.765564437 \\ 2.265564437 \end{bmatrix}$$

$$\langle x, y\rangle = \left[\frac{\sqrt{10+4\sqrt{22}}}{6}, -\frac{1}{3}+\frac{\sqrt{22}}{6}\right]$$

$$f(x, y) = \frac{\left(10+4\sqrt{22}\right)\left(\frac{2}{3}+\frac{\sqrt{22}}{6}\right)}{36} - \frac{1}{6} + \frac{\sqrt{22}}{12}$$

$$\begin{bmatrix} 2.470805780 \\ -3.749745493 \\ -1.617865541 \end{bmatrix}$$

$$\langle x, y\rangle = \left[-\frac{\sqrt{10+4\sqrt{22}}}{6}, -\frac{1}{3}+\frac{\sqrt{22}}{6}\right]$$

$$f(x, y) = \frac{\left(10+4\sqrt{22}\right)\left(\frac{2}{3}+\frac{\sqrt{22}}{6}\right)}{36} - \frac{1}{6} + \frac{\sqrt{22}}{12}$$

$$\begin{bmatrix} 2.470805780 \\ -3.749745493 \\ -1.617865541 \end{bmatrix}$$

$$\langle x, y\rangle = \left[\frac{I}{6}\sqrt{-10+4\sqrt{22}}, -\frac{1}{3}-\frac{\sqrt{22}}{6}\right]$$

"Komplexe kritische Stelle ignoriert"

$$\langle x, y\rangle = \left[-\frac{I}{6}\sqrt{-10+4\sqrt{22}}, -\frac{1}{3}-\frac{\sqrt{22}}{6}\right]$$

"Komplexe kritische Stelle ignoriert" **(5.9)**

```
> # Hmm, alle indefinit ... dann optisch ablesen
> # Filtere komplexe Lösungen
> kritische_punkte := seq(kritische_punkte[ll], ll = 1..4):
> kp3d := [ seq(subs(kr, [ x, y, f(x, y) ]), kr in
  kritische_punkte) ];
> pp := pointplot3d(kp3d, symbol = circle,  symbolsize = 50,
  color = yellow);
> display([ pf, pf_constr, pp ]);
```

$$kp3d := \left[\left[0, 1, \frac{1}{2}\right], \left[0, -1, -\frac{1}{2}\right], \left[\frac{\sqrt{10+4\sqrt{22}}}{6}, -\frac{1}{3}+\frac{\sqrt{22}}{6},\right.\right.$$

$$\frac{\left(10+4\sqrt{22}\right)\left(\frac{2}{3}+\frac{\sqrt{22}}{6}\right)}{36}-\frac{1}{6}+\frac{\sqrt{22}}{12}\Bigg]\Bigg],\Bigg[\Bigg[-\frac{\sqrt{10+4\sqrt{22}}}{6},-\frac{1}{3}$$

$$+\frac{\sqrt{22}}{6},\frac{\left(10+4\sqrt{22}\right)\left(\frac{2}{3}+\frac{\sqrt{22}}{6}\right)}{36}-\frac{1}{6}+\frac{\sqrt{22}}{12}\Bigg]\Bigg]$$

$$pp := PLOT3D(\ldots)$$



Alternative: 1D Funktion

```
> h := t -> (f@param)(t);
> dh := D(h);
> d2h := D(dh);
```

$$h := t \mapsto (f@param)(t)$$

$$dh := t \mapsto -2\cos(t)\,(\sin(t)+1)\sin(t)+\cos(t)^3+\frac{\cos(t)}{2}$$

$$d2h := t \mapsto 2\sin(t)^2\,(\sin(t)+1)-5\cos(t)^2\sin(t)-2\cos(t)^2\,(\sin(t)+1) \qquad \textbf{(5.10)}$$

$$-\frac{\sin(t)}{2}$$

```
> # h(0), dh(0), d2h(0);
> kritische_punkte := solve({ dh(t) = 0 }, { t });
```

$$kritische\_punkte := \left\{t = \arctan\left(\frac{6\left(-\frac{1}{3}+\frac{\sqrt{22}}{6}\right)}{\sqrt{10+4\sqrt{22}}}\right)\right\}, \left\{t = -\arctan\left(\frac{6\left(-\frac{1}{3}+\frac{\sqrt{22}}{6}\right)}{\sqrt{10+4\sqrt{22}}}\right)+\pi\right\}, \left\{t = \arctan\left(-\frac{1}{3}-\frac{\sqrt{22}}{6}, \frac{\sqrt{10-4\sqrt{22}}}{6}\right)\right\}, \left\{t = \arctan\left(-\frac{1}{3}-\frac{\sqrt{22}}{6}, -\frac{\sqrt{10-4\sqrt{22}}}{6}\right)\right\}, \left\{t = \frac{\pi}{2}\right\} \quad \textbf{(5.11)}$$

```
> # Maple findet die kritische Stelle bei t = -Pi/2 nicht, fügen
  wir diese hinzu ...
> dh(-Pi/2);
> kritische_punkte := kritische_punkte, { t = -Pi/2 };
```

$$0$$

$$kritische\_punkte := \left\{t = \arctan\left(\frac{6\left(-\frac{1}{3}+\frac{\sqrt{22}}{6}\right)}{\sqrt{10+4\sqrt{22}}}\right)\right\}, \left\{t = -\arctan\left(\frac{6\left(-\frac{1}{3}+\frac{\sqrt{22}}{6}\right)}{\sqrt{10+4\sqrt{22}}}\right)+\pi\right\}, \left\{t = \arctan\left(-\frac{1}{3}-\frac{\sqrt{22}}{6}, \frac{\sqrt{10-4\sqrt{22}}}{6}\right)\right\}, \left\{t = \arctan\left(-\frac{1}{3}-\frac{\sqrt{22}}{6}, -\frac{\sqrt{10-4\sqrt{22}}}{6}\right)\right\}, \left\{t = \frac{\pi}{2}\right\}, \left\{t = -\frac{\pi}{2}\right\} \quad \textbf{(5.12)}$$

```
> # Oder vielleicht doch dabei?!?
> solve({ dh(t) = 0 }, { t }, AllSolutions=true) ;
> about(_Z1); about(_Z2);
```

$$\left\{t = \arctan\left(\frac{6\left(-\frac{1}{3}+\frac{\sqrt{22}}{6}\right)}{\sqrt{10+4\sqrt{22}}}\right)+2\pi\_Z2\sim\right\}, \left\{t = -\arctan\left(\frac{6\left(-\frac{1}{3}+\frac{\sqrt{22}}{6}\right)}{\sqrt{10+4\sqrt{22}}}\right)+\pi+2\pi\_Z2\sim\right\}, \left\{t = \arctan\left(-\frac{1}{3}-\frac{\sqrt{22}}{6}, \frac{\sqrt{10-4\sqrt{22}}}{6}\right)+2\pi\_Z2\sim\right\},$$

$$\left\{t = \arctan\left(-\frac{1}{3} - \frac{\sqrt{22}}{6}, -\frac{\sqrt{10 - 4\sqrt{22}}}{6}\right) + 2\pi\_Z2\sim\right\}, \left\{t = \frac{1}{2}\pi + \pi\_Z1\sim\right\}$$

Originally _Z1, renamed _Z1~:
  is assumed to be: integer

Originally _Z2, renamed _Z2~:
  is assumed to be: integer

```
> # Aha, die letzte Lösung ist Pi-periodisch, die anderen alle 2*
  Pi-periodisch.
> # Da unsere Parametrisierung 2*Pi-periodisch ist, sind es
  tatsächlich zwei verschiedene Lösungen!
> # (d.h. man muss die zweite Lösung von Hand zu obiger Liste
  hinzufügen)
> minMax[-1] := "Maximum":
> minMax[1] := "Minimum":
> minMax[0] := "???": # Bei 2. Ableitung = 0 müssen weitere
  Kriterien hinzugenommen werden
> for kr in kritische_punkte do
    #kr;
    xy := simplify(subs(kr, [param(t)])):
    print('t' = simplify(subs(kr, t)));
    print('<x, y>' = xy);
    if (not is(xy[1], real)) or
       (not is(xy[2], real)) then
     # Komplexe Werte überspringen
     print("Komplexe kritische Stelle ignoriert");
      next;
    end if;
    print('f(x, y)' = f(xy[1], xy[2]));
    # Kriterium 2. Ordnung
    d2h_val := simplify(subs(kr, d2h(t)));
    print('diff(f@param, t)'('t') = d2h_val);
    #typ := minMax[sign(d2h_val)];
    #print(typ);
  end do:
```

$$t = \arctan\left(\frac{-2 + \sqrt{22}}{\sqrt{10 + 4\sqrt{22}}}\right)$$

$$\langle x, y\rangle = \left[\frac{\sqrt{2}\sqrt{5 + 2\sqrt{2}\sqrt{11}}}{6}, -\frac{1}{3} + \frac{\sqrt{2}\sqrt{11}}{6}\right]$$

$$f(x, y) = \left(\frac{5}{18} + \frac{\sqrt{2}\sqrt{11}}{9}\right)\left(\frac{2}{3} + \frac{\sqrt{2}\sqrt{11}}{6}\right) - \frac{1}{6} + \frac{\sqrt{2}\sqrt{11}}{12}$$

$$\left(\frac{\partial}{\partial t}\,(f@param)\right)(t) = -\frac{\left(22\sqrt{2}+5\sqrt{11}\right)\sqrt{2}}{18}$$

$$t = -\arctan\left(\frac{-2+\sqrt{22}}{\sqrt{10+4\sqrt{22}}}\right)+\pi$$

$$\langle x, y\rangle = \left[-\frac{\sqrt{2}\sqrt{5+2\sqrt{2}\sqrt{11}}}{6}, -\frac{1}{3}+\frac{\sqrt{2}\sqrt{11}}{6}\right]$$

$$f(x, y) = \left(\frac{5}{18}+\frac{\sqrt{2}\sqrt{11}}{9}\right)\left(\frac{2}{3}+\frac{\sqrt{2}\sqrt{11}}{6}\right)-\frac{1}{6}+\frac{\sqrt{2}\sqrt{11}}{12}$$

$$\left(\frac{\partial}{\partial t}\,(f@param)\right)(t) = -\frac{\left(22\sqrt{2}+5\sqrt{11}\right)\sqrt{2}}{18}$$

$$t = \arctan\left(-\frac{1}{3}-\frac{\sqrt{22}}{6}, \frac{I}{6}\sqrt{-10+4\sqrt{22}}\right)$$

$$\langle x, y\rangle = \left[\frac{I}{6}\sqrt{-10+4\sqrt{22}}, -\frac{1}{3}-\frac{\sqrt{22}}{6}\right]$$

"Komplexe kritische Stelle ignoriert"

$$t = \arctan\left(-\frac{1}{3}-\frac{\sqrt{22}}{6}, -\frac{I}{6}\sqrt{-10+4\sqrt{22}}\right)$$

$$\langle x, y\rangle = \left[-\frac{I}{6}\sqrt{-10+4\sqrt{22}}, -\frac{1}{3}-\frac{\sqrt{22}}{6}\right]$$

"Komplexe kritische Stelle ignoriert"

$$t = \frac{\pi}{2}$$

$$\langle x, y\rangle = [0, 1]$$

$$f(x, y) = \frac{1}{2}$$

$$\left(\frac{\partial}{\partial t}\,(f@param)\right)(t) = \frac{7}{2}$$

$$t = -\frac{\pi}{2}$$

$$\langle x, y\rangle = [0, -1]$$

$$f(x, y) = -\frac{1}{2}$$

$$\left(\frac{\partial}{\partial t}\,(f@param)\right)(t) = \frac{1}{2} \qquad \textbf{(5.13)}$$

```
> # Und hier auch noch einmal zeichnen:
> kritische_punkte := seq(kritische_punkte[ll], ll in [ 1, 2, 5,
  6 ]);
```

$$kritische\_punkte := \left\{t = \arctan\left(\frac{6\left(-\frac{1}{3}+\frac{\sqrt{22}}{6}\right)}{\sqrt{10+4\sqrt{22}}}\right)\right\}, \left\{t = -\arctan\left(\frac{6\left(-\frac{1}{3}+\frac{\sqrt{22}}{6}\right)}{\sqrt{10+4\sqrt{22}}}\right)+\pi\right\}, \left\{t = \frac{\pi}{2}\right\}, \left\{t = -\frac{\pi}{2}\right\} \quad (5.14)$$

```
> kp3d := [ seq(subs(kr, [ param(t), (f@param)(t) ]), kr in
  kritische_punkte) ]:
> pp := pointplot3d(kp3d, symbol = circle,  symbolsize = 50,
  color = yellow);
> display([ pf, pf_constr, pp ]);
```

$$pp := PLOT3D(...)$$



Weitere Alternative: Lass Maple eine (stückweise) Parametrisierung finden

```
> solve(g(x, y) = 0, x);
```

$$\sqrt{-y^2+1}, -\sqrt{-y^2+1} \quad (5.15)$$

```
> # Dann die beiden Lösungen als Parametrisierungen benutzen.
```